

## 5.7. Exercise

**P5.1** In the beam contact problem in Section 5.2.1, determine the contact force and tip deflection using the Lagrange multiplier method. Choose the gap  $g$  as a Lagrange multiplier.

### Solution:

The contact consistency condition is given as

$$\lambda g = \lambda(v_{\text{tip}} - \delta) = 0$$

Since the gap is considered as a Lagrange multiplier, the contact force,  $\lambda$ , needs to be written in terms of the gap, or equivalently, in terms of  $v_{\text{tip}}$ . From the tip deflection formula in Eq. (5.1),

$$\lambda = \frac{3}{8} \times 10^3 - 3 \times 10^5 \times v_{\text{tip}}$$

By substituting  $v_{\text{tip}}$  into the consistency condition in Eq. (5.3), we have

$$\left( \frac{3}{8} \times 10^3 - 3 \times 10^5 \times v_{\text{tip}} \right) (v_{\text{tip}} - 1 \times 10^{-3}) = 0$$

The above equation has two solutions:  $v_{\text{tip}} = 1.25 \times 10^{-3}$  and  $v_{\text{tip}} = 1 \times 10^{-3}$ . The former yields  $\lambda = 0$  but  $g = 0.25 \times 10^{-3} > 0$ , which violates the contact requirement. Therefore, it is invalid. The latter, however, yields  $\lambda = 75\text{N} > 0$  and  $g = 0$ , which satisfies the contact requirement. This means that the beam contacts with the rigid block with the contact force of 75N. ■

---

**P5.2** For the beam contact problem in Section 5.2.1, determine the contact force and tip deflection using the Lagrange multiplier method. Model the beam using a two-node Euler beam element. Compare the results with the results in Section 5.2.1, and explain the reason for different results.

### Solution:

The finite element equation for a two-node Euler beam element with a distributed load and clamped condition at Node 1 becomes

$$\frac{EI}{L^3} \begin{bmatrix} 12 & 6L & -12 & 6L \\ 6L & 4L^2 & -6L & 2L^2 \\ -12 & -6L & 12 & -6L \\ 6L & 2L^2 & -6L & 4L^2 \end{bmatrix} \begin{bmatrix} v_1 \\ \theta_1 \\ v_2 \\ \theta_2 \end{bmatrix} = \begin{bmatrix} -qL / 2 + R_1 \\ -qL^2 / 12 + C_1 \\ -qL / 2 \\ qL^2 / 12 \end{bmatrix}$$

In addition, since the first two degrees-of-freedom are fixed, we can delete the first two rows and columns, yielding the following reduced equation:

$$\frac{EI}{L^3} \begin{bmatrix} 12 & -6L \\ -6L & 4L^2 \end{bmatrix} \begin{Bmatrix} v_2 \\ \theta_2 \end{Bmatrix} = \begin{Bmatrix} -qL / 2 \\ qL^2 / 12 \end{Bmatrix}$$

In the Lagrange multiple method, it is considered that the Lagrange multiplier is the contact force in the interface. Therefore, the above equation is modified as

$$10^5 \begin{bmatrix} 12 & -6 \\ -6 & 4 \end{bmatrix} \begin{Bmatrix} v_2 \\ \theta_2 \end{Bmatrix} = \begin{Bmatrix} -500 + \lambda \\ 1000 / 12 \end{Bmatrix}$$

Solving the above equation yields the vertical tip displacement as a function of Lagrange multiplier as

$$v_2 = \frac{10^{-5}}{3} \lambda - 0.00125$$

Using this relation, the contact consistency condition in Eq. (5.3) can be written as

$$\lambda g = \lambda(v_2 + \delta) = \lambda \left( \frac{10^{-5}}{3} \lambda - 0.00025 \right) = 0$$

Different from Eq. (5.2),  $g = v_2 + \delta$  is used because the nodal displacement is positive in the positive y-direction. The above quadratic equations has two solutions:  $\lambda = 0\text{N}$  and  $\lambda = 75\text{N}$ . When  $\lambda = 0\text{N}$ , the gap becomes  $g = 0.00025 > 0$ , which violates the condition of no penetration. Therefore, this cannot be a possible configuration. On the other hand, when  $\lambda = 75\text{N}$ , the gap becomes  $g = 0$ . Since this solution satisfies all requirements, this is the solution. In fact, the solution is consistent with the solution from Section 5.2.1. ■

**P5.3** For the frictional contact problem in Section 5.2.2, determine the frictional force and slip displacement using the Lagrange multiplier method. Choose the slip  $u_{\text{tip}}$  as a Lagrange multiplier.

**Solution:**

In the Lagrange multiplier method, the consistency condition is used to find the unknown slip displacement. Using the tip displacement formula, the consistency condition can be written as

$$\frac{(P - t)L}{EA} (t - \mu\lambda) = 0$$


The above consistency condition has two solutions:  $t = P$  and  $t = \mu\lambda$ . The condition of  $t = P$  is equivalent to the requirement of  $u_{\text{tip}} = 0$ . Therefore, the other condition of  $t - \mu\lambda < 0$  should be checked. This condition is satisfied if  $P < \mu\lambda$ , which corresponds to the stick condition. For the given load conditions, however,  $P > \mu\lambda$ , and this cannot satisfy the requirement.

The condition of  $t = \mu\lambda$  corresponds to the slip condition, and needs to show that  $u_{\text{tip}} > 0$ . The tip displacement can then be expressed by

$$u_{\text{tip}} = \frac{(P - \mu\lambda)L}{EA} = 0.625\text{mm} > 0$$

Since the tip displacement is greater than zero, it satisfies the requirement of the slip condition, and it is a valid state. ■

**P5.4** During a Newton-Raphson iteration, a rectangular plane element is in contact with a rigid surface as shown in the figure. Due to the penalty method, the penetration of  $g = -1 \times 10^{-4}\text{m}$  is observed with penalty parameter  $\omega_n = 10^6$ . In the two-dimensional problem, the element has eight degrees-of-freedom  $\{u_{1x}, u_{1y}, u_{2x}, u_{2y}, u_{3x}, u_{3y}, u_{4x}, u_{4y}\}^T$ . Calculate the contact force and contact stiffness matrix in terms of 8x1 vector and 8x8 matrix, respectively.



Rigid surface Figure P5.4 Contact of a rectangular block

**Solution:**

Since the element boundary is straight and the rigid surface is flat, the contact boundary is constant, the edge between Nodes 1 and 2. The contact force term in Eq. (5.59) can be applicable for both Nodes 1 and 2. Using  $\mathbf{e}_n = [0, 1]^T$ , the contact forces are

$$\mathbf{f}_{n1}^c = \mathbf{f}_{n2}^c = -\omega_n g_n \mathbf{e}_n = \{0, 10^2\}^T$$

Therefore, through the assembly process, the contact force vector of the element becomes

$$\mathbf{F}_n^c = \{0 \quad 10^2 \quad 0 \quad 10^2 \quad 0 \quad 0 \quad 0 \quad 0\}^T$$

This is equivalent to applying vertical forces at Nodes 1 and 2.

The contact stiffness is the same for both nodes, as

$$\mathbf{k}_{n1}^c = \mathbf{k}_{n2}^c = \omega_n \mathbf{e}_n \mathbf{e}_n^T = \begin{bmatrix} 0 & 0 \\ 0 & \omega_n \end{bmatrix}$$

Therefore, after assembly, the element contact stiffness matrix becomes

$$\mathbf{K}_n^c = \begin{bmatrix} 0 & & & & & & \\ & \omega_n & & & & & 0 \\ & & 0 & & & & \\ & & & \omega_n & & & \\ & & & & 0 & & \\ & & & & & 0 & \\ 0 & & & & & & 0 \\ & & & & & & & 0 \end{bmatrix}$$



**P5.5** A sphere of radius  $r = 8\text{mm}$  is pressed against a rigid flat plane. Using a commercial program, determine the contact radius,  $a$ , for a given load  $F = (30 \times 2\pi)\text{N}$ . Assume a linear elastic material with Young's modulus  $E = 1000 \text{ N/mm}^2$  and Poisson's ratio  $\nu = 0.3$ . Use an axi-symmetric model. Compare the finite element result with the analytical contact radius of  $a = 1.010\text{mm}$ .





Figure P5.5 Contact of a sphere

**Solution:**

The problem can be solved using different commercial programs, but ANSYS is used to solve it. An axisymmetric model is used. A node is placed near the expected radius of contact. Midside nodes are removed along the surface where contact is likely to occur. The model is comprised of both PLANE82 and PLANE183. The model is solved using 3-D node-to-node contact elements (CONTA178). The following figure shows finite element mesh along with boundary conditions:



As shown in the following table, the contact radius 1.011mm is close to the target value of 1.010mm.

	Target	ANSYS	Ratio
A, mm	1.010	1.011	1.001

The following program is the ANSYS command script to solve the problem:

```

/PREP7
SMRT,OFF
/TITLE, STATIC HERTZ CONTACT PROBLEM SOLVED USING CONTAC178 ELEMENTS
ET,1,PLANE82,,,1      ! AXISYMMETRIC ELEMENTS
ET,2,PLANE183,,,1
ET,3,CONTA178,,4      ! NODAL CONTACT
R,1
RMOD,1,7,1            !CONTACT NORMAL ALONG UY
MP,EX,1,1E3
MP,NUXY,1,.3
LOCAL,11,1,0,8,0      ! LOCAL CYLINDRICAL C.S. AT CENTERLINE
K,1,8,-90              ! DEFINE KEYPOINTS
K,2,8
K,3,7.5,-90
K,4,7.5
K,5
K,6,8,-82.65          ! PLACE KEYPOINT AND NODE AT EXPECTED CONTACT RADIUS
K,7,7.5,-82.65
L,1,3                  ! DEFINE LINES
L,2,4
L,6,7
LESIZE,ALL,,,1        ! DEFINE ELEMENT DIVISIONS ON ALL EXISTING LINES
A,1,6,7,3              ! DEFINE AREAS
A,6,2,4,7
A,3,7,4,5
LOCAL,12,0,0,8,0
ARSYM,Y,1,3,1         ! CREATE HALF-SYMMETRY MODEL
NUMMRG,KPOI
ESIZE,,4               ! DEFINE ELEMENT DIVISIONS ON REMAINING LINES
LESIZE,4,,,5
*REPEAT,2,1
LESIZE,6,,,8,8
LESIZE,7,,,8,(1/8)
LESIZE,10,,,1
*REPEAT,2,2
LESIZE,9,,,6,.2
TYPE,1                 ! CREATE NODES AND ELEMENTS
AMESH,1,2,1
AMESH,4,5,1
TYPE,2
MSHAPE,1,2D
MSHKEY,0
AMESH,3,6,3
CSYS,0
N,1001,-1,1E-8        !NODE 1001 IS THE GROUND
D,1001,ALL             !X POSITION DOES NOT MATTER IN THIS CASE BECAUSE
                       !THE CONTACT NORMAL IS ONLY ALONG UY

TYPE,3
REAL,1
EN,205,1001,2          !USE THE SAME ELEMENT NUMBERS AS VM63 FOR POST-PROC
EN,201,1001,4
EN,202,1001,6
EN,203,1001,8
EN,204,1001,10
EN,206,1001,31

MODMSH,NOCHECK
TYPE,1                 ! REMOVE MIDSIDE NODES ALONG CONTACT SURFACE
EMODIF,7,7,0
*REPEAT,6,1
MODMSH,CHECK

```

```

FINISH
/SOLU
NSEL,S,LOC,X,-.01,.01    ! BOUNDARY CONDITIONS AND LOADING
D,ALL,UX,0
NSEL,R,LOC,Y,0
D,ALL,UY,0
NSEL,ALL
LOAD=0
*CREATE,LOADSTEP          ! MACRO TO INCREMENTALLY APPLY LOAD
FK,8,FY,ARG1
SOLVE
*END
*DO,I,1,3
  LOAD=LOAD-10
  *USE,LOADSTEP,LOAD*6.2831853
*ENDDO
FINISH
/POST1                    ! POSTPROCESS
/OUT,
SET,3
ESEL,,TYPE,,3
ETABLE,RFOR,SMISC,1
NSLE
PRETAB,RFOR              ! PRINT REACTION FORCE TO DETERMINE CONTACT AREA
SSUM                    ! SUM OF REACTION FORCE
NLIST                   ! LIST COORDINATES OF NODES OF CONTACT SURFACE
PRNSOL,U,COMP           ! LIST DISPLACEMENTS OF NODES
/COM  CALCULATE RATIO OF A - ACTUAL TO A - TARGET
PI=(4*ATAN(1))
LOAD=- (LOAD)*(2*PI)
ATAR=(0.88*((LOAD*0.008)**(1/3)))    ! A - TARGET
*GET,EMAX,ELEM,,NUM,MAX
*DO,ENUM,201,EMAX        ! START SEARCH FROM ELEM 201
  *GET,GRFR,ELEM,ENUM,ETAB,RFOR      ! FIND LAST ELEMENT IN CONTACT
  *IF,GRFR,EQ,0.0,EXIT
*ENDDO
ESEL,,ELEM,,(ENUM-1)    ! SELECT LAST CONTACTING ELEMENT
NSLE                    ! SELECT NODES ATTACHED TO SELECTED ELEMENTS
*GET,NMIN,NODE,0,NUM,MIN
NODX=NX(NMIN)
NODY=NY(NMIN)
NUX =UX(NMIN)
NUY =UY(NMIN)
AACT=NODX+NUX           ! A - ACTUAL
YCHK=NODY+NUY
RATA=(AACT/ATAR)        ! RATIO
*status,parm
*DIM,LABEL,CHAR,1,2
*DIM,VALUE,,1,3
LABEL(1,1) = '          A,'
LABEL(1,2) = ' mm      '
*VFILL,VALUE(1,1),DATA,1.010
*VFILL,VALUE(1,2),DATA,AACT
*VFILL,VALUE(1,3),DATA,ABS(AACT/1.010)
/COM
/COM,-----
/COM,----- RESULTS COMPARISON (OBTAINED USING CONTACT178 ELEMENTS)-----
/COM,
/COM,          |   TARGET   |   ANSYS   |   RATIO
/COM,
*VWRITE,LABEL(1,1),LABEL(1,2),VALUE(1,1),VALUE(1,2),VALUE(1,3)
(1X,A8,A8,'      ',F10.3,'      ',F10.3,'      ',1F5.3)
/COM,-----

```

/OUT  
FINISH



**P5.6** A long rubber cylinder with radius  $r = 200\text{mm}$  is pressed between two rigid plates using a maximum imposed displacement of  $\delta_{\max} = 200\text{mm}$ . Determine the force-deflection response. Use Mooney-Rivlin material with  $A_{10} = 0.293\text{MPa}$  and  $A_{01} = 0.177\text{MPa}$ . Assume a plane strain condition and symmetry. Compare the results with the target results of  $F = 250\text{N}$  at  $\delta = 100\text{mm}$  and  $F = 1400\text{N}$  at  $\delta = 200\text{mm}$ .





Figure P5.6 Rubber cylinder contact problem

**Solution:**



The problem can be solved using different commercial programs, but ANSYS is used to solve it. Since the rubber is in the cylindrical shape, a plane strain solution is assumed. Due to geometric and loading symmetry, the analysis can be performed using one quarter of the cross section, as shown in the figure below.



All nodes on the left edge ( $x = 0$ ) are constrained,  $UX = 0$ . All nodes on the top edge ( $y = 0$ ) are coupled in  $UY$ . An imposed displacement of  $-0.1\text{m}$  acts upon the coupled nodes. Using ANSYS, the problem can be solved in several ways:

- 1) A 2-D model using PLANE182 with CONTA175 elements
- 2) A 3-D model using SOLID185 with CONTA175 elements
- 3) A 2-D model using PLANE182 with CONTA175 element and solved using Lagrange Multipliers method.
- 4) A 3-D model using SOLID185 with CONTA175 element and solved using Lagrange Multipliers method.

In the 3-D case, a MESH200 element is used as the target face for the automatic generation of contact elements. The target surface is given a high contact stiffness ( $KN = 2000\text{ MPa}$ ) to model a rigid surface and no contact stiffness is required to be specified while performing the solution using Lagrange Multipliers method. The following two figures show deformed geometry and force-displacement curve.



The force results at  $d = 100\text{mm}$  and  $d = 200\text{mm}$  are compared with the values in the literature (T. Tussman, K-J Bathe, "A Finite Element Formulation for Nonlinear Incompressible Elastic and Inelastic Analysis", pg. 385, fig. 6.14). The following table compares the results with different combinations of elements. It is found that the numerical simulation has an error less than 6.6%.

	Target	ANSYS	Ratio
2-D with augmented Lagrangian method			
Force at $d = 100\text{mm}$	250.00	266.05	1.064
Force at $d = 200\text{mm}$	1400.00	1397.06	0.998
2-D with augmented Lagrangian method			
Force at $d = 100\text{mm}$	250.00	258.77	1.035
Force at $d = 200\text{mm}$	1400.00	1398.79	0.999
2-D with Lagrange multiplier method			
Force at $d = 100\text{mm}$	250.00	266.20	1.065
Force at $d = 200\text{mm}$	1400.00	1400.42	1.000
3-D with Lagrange multiplier method			
Force at $d = 100\text{mm}$	250.00	266.40	1.066
Force at $d = 200\text{mm}$	1400.00	1400.48	1.000



The following program is the ANSYS command script to solve the problem:

```

R = 200                                ! RADIUS OF CYLINDER (mm)
/PREP7
smrt,off
/TITLE, RUBBER CYLINDER PRESSED BETWEEN TWO PLATES
ET,1,PLANE182, , ,2                    ! 2-D PLANE-STRAIN 4-NODE STRUCTURAL SOLID
KEYOPT,1,6,1
ET,2,CONTA175                          ! 2-D 1-NODE NODE-TO-SURFACE CONTACT ELEMENT
R,2, , , -2000                         ! SET SURFACE STIFFNESS
ET,3,TARGE169                          ! 2-D TARGET ELEMENT
MP,EX,1,2.82                           ! YOUNG'S MODULUS [MPa]
MP,NUXY,1,0.49967                      ! POISSON'S RATIO
C10 = 0.293
C01 = 0.177
NU1 = 0.49967
DD = (1-2*NU1)/(C10+C01)
TB,HYPER,1,1,2,MOONEY
TBDATA,1,C10,C01,DD
CSYS,1                                ! SWITCH TO CYLINDRICAL C.S.
K,1                                    ! DEFINE KEYPOINTS
K,2,R,-90
K,3,R
K,4,(0.5*R),-90
K,5,(0.6*R),-45
K,6,(0.5*R)
K,7,R,-45
L,2,7
L,7,3
CSYS,0                                ! SWITCH TO CARTESIAN C.S.
A,2,7,5,4
A,7,3,6,5
A,4,5,6,1
ESIZE, ,4                             ! SET ELEMENT DIVISION SIZE
AMESH,ALL                             ! MESH ALL AREAS
SAVE                                   ! SAVE MODEL FOR MORE ANALYSIS
N,1001,(-2*R),-R                       ! TARGET SURFACE NODES
N,1002,(2*R),-R
NSSEL,S,NODE,,1001,1002
TYPE,3
REAL,2
TSHAP,LINE                             ! SET TARGET SHAPE TO LINE
E,1002,1001                             ! GENERATE RIGID TARGET
D,ALL,ALL,0                             ! FIX TARGET
NSSEL,S,LOC,X                           ! SELECT LEFT EDGE
D,ALL,UX                                ! CONSTRAIN LEFT EDGE IN UX
NSSEL,S,LOC,Y                           ! SELECT TOP EDGE
CP,1,UY,ALL                             ! COUPLE TOP EDGE IN UY
*GET,NCEN,NODE,,NUM,MIN                 ! GET MINIMUM NODE NUMBER FROM SELECTED SET
NSSEL,ALL
CSYS,1                                ! SWITCH TO CYLINDRICAL C.S.
ESEL,S,TYPE,,1
NSLE
NSSEL,R,LOC,X,R
TYPE,2
REAL,2
ESURF                                  ! DEFINE CONTACT ELEMENTS
ALLSEL,ALL
CSYS,0
SAVE,CONT2D                             ! SAVE 2D CONTACT MODEL FOR SECOND ANALYSIS
FINISH

*CREATE,SOLVIT,MAC                       ! MACRO TO SOLVE MODEL

```

4-10

```

/SOLU
ANTYPE,STATIC
CNVTOL,F,,,,-1
NLGEOM,ON                                ! INCLUDE LARGE DEFORMATION EFFECTS
NSUBST,6                                ! SPECIFY NUMBER OF SUBSTEPS IN LOAD STEP
OUTRES,,1                                ! WRITE SOLUTION FOR EVERY SUBSTEP
D,NCEN,UY,-100                           ! APPLY DISPLACEMENT UY = -100 TO COUPLED NODES
SOLVE
FINISH
*END
SOLVIT                                    ! USE MACRO SOLVIT

*CREATE,PLOTS,MAC                         ! MACRO FOR POST-PROCESSING
/POST1
/DSCALE,1,1
PLDISP,1                                ! PLOT DISPLACED SHAPE
FINISH
/POST26
/AXLAB,Y,FORCE
/AXLAB,X,DISPLACEMENT
NSOL,2,NCEN,U,Y
RFORCE,3,NCEN,F,Y
PROD,2,2,,,,,,,,-2
PROD,3,3,,,,,,,,-2
XVAR,2
PLVAR,3                                ! PLOT DISPLACEMENT VS FORCE
PRVAR,2,3                               ! PRINT DISPLACEMENT, FORCE
*GET,F1,VARI,3,RTIME,.5
*GET,F2,VARI,3,RTIME,1
*DIM,LABEL,CHAR,2,2
*DIM,VALUE,,2,3
LABEL(1,1) = 'F (N) @ ','F (N) @ '
LABEL(1,2) = '.1','.2'
*VFILL,VALUE(1,1),DATA,250,1400
*VFILL,VALUE(1,2),DATA,F1,F2
*VFILL,VALUE(1,3),DATA,ABS(F1/250),ABS(F2/1400)
FINISH
*END
PLOTS                                    ! USE MACRO PLOTS
SAVE,TABLE_1

RESUME
/PREP7
SMRT,OFF
ET,5,SOLID185                            ! 3-D 8-NODE STRUCTURAL SOLID
KEYOPT,5,6,1
ET,6,CONTA175                            ! 3-D 1-NODE NODE-TO-SURFACE CONTACT ELEMENT
R,6,,,-2000,-0.1,
ET,7,TARGE170                            ! 3-D TARGET ELEMENT
ET,8,MESH200,6                           ! 2-D 4-NODED QUAD
R,8,0.05
ALLSEL
TYPE,5
ESIZE,,1
VEXT,ALL,,,,1
N,1001,,-R                                ! CREATE TARGET PLANE OF NODES
N,1002,2*R,-R
N,1003,2*R,-R,8*R
N,1004,,-R,8*R
TYPE,8
REAL,8
E,1002,1001,1004,1003
NSEL,S,NODE,,1001,1004

```

```

TYPE,7
REAL,6
ESURF                                ! GENERATE TARGET ELEMENTS
D,ALL,ALL,0
CSYS,1                                ! SWITCH TO CYLINDRICAL C.S.
ESEL,S,TYPE,,5
NSLE
NSEL,R,LOC,X,R
ESEL,S,TYPE,,5,7
TYPE,6
REAL,6
ESURF
CSYS,0                                ! SWITCH TO CARTESIAN C.S.
NSEL,ALL
D,ALL,UZ                                ! CONSTRAIN ALL NODES IN Z (PLANE STRAIN)
NSEL,S,LOC,X
D,ALL,UX
NSEL,S,LOC,Y
CP,1,UY,ALL                            ! COUPLE TOP NODES IN Y
*GET,NCEN,NODE,,NUM,MIN
ESEL,S,TYPE,,5,7
NSLE
SAVE,CONT3D                            ! SAVE 3D CONTACT MODEL FOR SECOND ANALYSIS
FINISH
SOLVIT                                ! USE MACRO TO OBTAIN SOLUTION
PLOTS                                 ! USE MACRO TO POSTPROCESS
SAVE,TABLE_2

RESUME,CONT2D                            ! RESUME CONT175 -2D MODEL
/PREP7
R,2                                    ! CONTACT STIFFNESS IS NOT REQUIRED
KEYOPT,2,2,3                            ! LAGRANGE MULTIPLIER ON CONTACT NORMAL AND
                                           ! PENALTY ON TANGENT

FINISH
SOLVIT                                ! USE MACRO TO OBTAIN SOLUTION
PLOTS                                 ! USE MACRO TO POSTPROCESS
SAVE,TABLE_3

RESUME,CONT3D                            ! RESUME CONT175 -3D MODEL
/PREP7
R,6                                    ! CONTACT STIFFNESS IS NOT REQUIRED
KEYOPT,6,2,4                            ! PURE LAGRANGE MULTIPLIER ON CONTACT NORMAL AND
                                           ! TANGENT

FINISH
SOLVIT                                ! USE MACRO TO OBTAIN SOLUTION
PLOTS                                 ! USE MACRO TO POSTPROCESS
SAVE,TABLE_4

RESUME,TABLE_1
/COM
/OUT,results,vrt
/COM,----- RESULTS COMPARISON -----
/COM,
/COM,          |   TARGET   |   ANSYS   |   RATIO
/COM,
/COM,RESULTS USING PLANE182 AND 2D-CONTA175:
*VWRITE,LABEL(1,1),LABEL(1,2),VALUE(1,1),VALUE(1,2),VALUE(1,3)
(1X,A8,A8,'    ',F10.2,'    ',F10.2,'    ',1F5.3)
/NOPR
RESUME,TABLE_2
/GOPR
/COM,
/COM,RESULTS USING SOLID185 AND 3D-CONTA175:

```

```

*VWRITE,LABEL(1,1),LABEL(1,2),VALUE(1,1),VALUE(1,2),VALUE(1,3)
(1X,A8,A8,' ',F10.2,' ',F10.2,' ',1F5.3)
/NOPR
RESUME, TABLE_3
/GOPR
/COM,
/COM, RESULTS USING PLANE182 AND 2D-CONTA175 WITH K(2)=3:
*VWRITE,LABEL(1,1),LABEL(1,2),VALUE(1,1),VALUE(1,2),VALUE(1,3)
(1X,A8,A8,' ',F10.2,' ',F10.2,' ',1F5.3)
/NOPR
RESUME, TABLE_4
/GOPR
/COM,
/COM, RESULTS USING SOLID185 AND 3D-CONTA175 WITH K(2)=4:
*VWRITE,LABEL(1,1),LABEL(1,2),VALUE(1,1),VALUE(1,2),VALUE(1,3)
(1X,A8,A8,' ',F10.2,' ',F10.2,' ',1F5.3)
/COM,-----
/COM,
/OUT
FINISH

```

**P5.7** Two long cylinders of radii  $R_1 = 10\text{mm}$  and  $R_2 = 13\text{mm}$ , in frictionless contact with their axes parallel to each other, are pressed together with a force per unit length,  $F = 3200\text{N/mm}$ . Determine the semi-contact length  $b$  and the approach distance  $d$ . Both materials are linear elastic with  $E_1 = 30000\text{N/mm}^2$  and  $\nu_1 = 0.25$  for Cylinder 1, and  $E_2 = 29120\text{N/mm}^2$  and  $\nu_2 = 0.3$  for Cylinder 2. Assume a plane stress condition with a unit thickness and symmetry. Compare the results with the target results of  $d = 0.4181\text{mm}$  and  $b = 1.20\text{mm}$ .




Figure P5.7 Hertzian contact problem

**Solution:**

The problem can be solved using different commercial programs, but ANSYS is used to solve it. Each analysis uses two load steps; in the first load step a small imposed displacement is used on the upper cylinder to engage contact, whereas in the second load step the imposed displacement is deleted and the force load is applied. The problem is solved in four different ways:

- 1) Contact Algorithm: Augmented Lagrangian - KEYOPT(2) = 0
  - 2-D analysis with PLANE182 and CONTA175
  - 3-D analysis with SOLID185 and CONTA175
- 2) Contact Algorithm: Lagrange Multiplier - KEYOPT(2) = 3
  - 2-D analysis with PLANE182 and CONTA175
  - 3-D analysis with SOLID185 and CONTA175

Plane stress condition is modeled using a unit thickness slice through the cylinders. The region modeled is shown shaded in the problem sketch. The ESURF command is used to automatically generate the contact and target elements between "contactor" nodes on the upper cylinder and "target" nodes on the lower cylinder. The default value of contact stiffness FKN is chosen while performing a solution using Augmented Lagrangian contact algorithm (KEYOPT(2) = 0) whereas no contact stiffness input is required to be specified while performing a solution using Lagrange Multiplier contact algorithm (KEYOPT(2) = 3).

The approach distance,  $d = -0.4181$ , and the radius of contact region,  $b = 1.20\text{mm}$  are compared with the values in the literature (N. Chandrasekaran, W. E. Haisler, R. E. Goforth, "Finite Element Analysis of Hertz Contact Problem with Friction", Finite Elements in Analysis and Design, Vol. 3, 1987, pp. 39-56.). The following table compares the results with different combinations of elements. It is found that the numerical simulation has an error less than 3.3%.

		Target	ANSYS	Ratio
Augmented Lagrangian method				
PLANE182	d (mm)	-0.4181	-0.4183	1.000
	b (mm)	1.20	1.1609	0.967
SOLID185	d (mm)	-0.481	-0.4191	1.002
	b (mm)	1.20	1.1609	0.967
Lagrange multiplier method				
PLANE182	d (mm)	-0.4181	-0.4181	1.000
	b (mm)	1.20	1.1609	0.967
SOLID185	d (mm)	-0.4181	-0.4190	1.002
	b (mm)	1.20	1.1609	0.967

The following program is the ANSYS command script to solve the problem:

```

/TITLE, HERTZ CONTACT BETWEEN TWO CYLINDERS
/COM
/COM  2-D ANALYSIS USING PLANE182 AND CONTA175
/COM  CONTACT ALGORITHM: AUGMENTED LAGRANGIAN - KEYOPT(2) = 0
/COM
/PREP7
SMRT,OFF
ANTYPE,STATIC
ET,1,PLANE182          ! 2-D SOLID ELEMENTS
ET,2,TARGE169         ! 2-D TARGET ELEMENTS
ET,3,CONTA175         ! 2-D CONTACT ELEMENTS
MP,EX,1,30000         ! SMALLER CYLINDER PROPERTIES
MP,NUXY,1,0.25
MP,EX,2,29120         ! LARGER CYLINDER PROPERTIES
MP,NUXY,2,0.30

```

4-14

```
CSYS,1
K,1                      ! CREATE BIGGER CYLINDER
K,2,13
K,3,13,82
K,4,13,90
K,5,11,90
L,1,5
L,2,3
LESIZE,ALL,,7
L,3,4                    ! TARGET SURFACE (LINE 3)
LOCAL,11,1,,13
L,3,5
CSYS,1
A,1,2,3,5
A,5,3,4,4
MAT,2
MSHK,1                   ! MAPPED AREA MESH
MSHA,0,2D                ! USING QUADS
ESIZE,,4
AMESH,1,2
LOCAL,12,1,,23-1E-5,, -90 ! INTRODUCE SLIGHT INTERFERENCE
K,11                      ! CREATE SMALLER CYLINDER
K,12,10
K,13,10,8
K,14,10,90
K,15,8
L,11,15
L,13,14
LESIZE,7,,6
LESIZE,8,,6
L,12,13                  ! CONTACT SURFACE (LINE 9)
CSYS,11
L,13,15
CSYS,12
MAT,1
A,12,13,15,15
A,15,13,14,11
ESIZE,,6
AMESH,3,4
LSEL,S,LINE,,9           ! SELECT CONTACT NODES ON SMALLER CYLINDER
NSLL,,1
CM,CYL1,NODE
REAL,1
TYPE,3
ESURF                    ! GENERATE COTAC175 ELEMENTS
LSEL,S,LINE,,3
NSLL,,1                  ! SELECT TARGET NODES ON BIGGER CYLINDER
REAL,1
TYPE,2
ESURF                    ! GENERATE TARGE169 ELEMENTS
NSEL,ALL
CSYS,0
NSEL,S,LOC,Y,23          ! SELECT TOP EDGE OF MODEL
CP,1,UY,ALL              ! COUPLE NODES ON TOP EDGE
*GET,NC,NODE,,NUM,MIN    ! GET LOWEST NODE NUMBER (MASTER)
NSEL,S,LOC,X              ! SYMMETRY CONSTRAINTS
D,ALL,UX
NSEL,S,LOC,Y
D,ALL,UY
NSEL,ALL
FINISH
SAVE,MODEL2D
```

```

*CREATE,SOLV2D,MAC      ! CREATE SOLUTION MACRO FOR 2-D CASE
/SOLU
D,NC,UY,-0.005         ! APPLY SMALL DISPLACEMENT TO ENGAGE CONTACT
SOLVE                  ! SOLVE FIRST LOAD STEP
DDELETE,NC,UY          ! DELETE IMPOSED DISPLACEMENT
F,NC,FY,-1600          ! APPLY HALF LOAD ON (SYMMETRY) MODEL
nsub,2,10,1
SOLVE                  ! SOLVE SECOND LOAD STEP
FINISH
*END

SOLV2D                  ! EXECUTE SOLUTION MACRO FOR 2-D CASE

*CREATE,RES2D,MAC       ! CREATE RESULTS MACRO FOR 2-D CASE
/POST1
NSEL,,LOC,Y,23         ! SELECT TOP EDGE OF SMALLER CYLINDER
*GET,D,NODE,NC,U,Y      ! GET APPROACH DISTANCE (D)
ESEL,S,TYPE,,3         ! SELECT CONTACT ELEMENTS
ETABLE,NSTAT,CONT,STAT ! STORE CONTACT STATUS
ESEL,R,ETAB,NSTAT,2,2   ! SELECT ELEMENTS WITH CONTACT (STAT=2)
CMSEL,S,CYL1           ! SELECT CONTACT COMPONENT NODES
NSLE,R                 ! RESELECT NODES WITH CONTACT
NSORT,LOC,X,1          ! SORT CONTACT NODES BY ASCENDING X LOCATION
*GET,B,SORT,,MAX        ! GET SEMI-CONTACT LENGTH (B)
*STATUS,PARM
*DIM,LABEL,CHAR,2,2
*DIM,VALUE,,2,3
LABEL(1,1) = 'AP DIS ', 'S-CON LEN '
LABEL(1,2) = ' mm', ' mm'
*VFILL,VALUE(1,1),DATA,-.4181,1.2
*VFILL,VALUE(1,2),DATA,D,B
*VFILL,VALUE(1,3),DATA,ABS(D/.4181),ABS(B/1.2)
FINISH
*END

RES2D                  ! EXECUTE POSTPROCESSING MACRO FOR 2-D CASE
SAVE,TABLE_1

/CLEAR, NOSTART
/COM
/COM 3-D ANALYSIS USING SOLID185 AND CONTA175
/COM CONTACT ALGORITHM: AUGMENTED LAGRANGIAN - KEYOPT(2) = 0
/COM
/PREP7 $SMRT,OFF
ANTYPE,STATIC
ET,1,SOLID185          ! 3-D SOLID ELEMENTS
ET,2,170               ! 3-D TARGET ELEMENTS
ET,3,175               ! 3-D CONTACT ELEMENTS
MP,EX,1,30000          ! SMALLER CYLINDER PROPERTIES
MP,NUXY,1,0.25
MP,EX,2,29120          ! LARGER CYLINDER PROPERTIES
MP,NUXY,2,0.30
CSYS,1
K,1                    ! CREATE LOWER BIGGER CYLINDER
K,2,13
K,3,13,82
K,4,13,90
K,5,11,90
KGEN,2,1,5,1,,,1,100 ! UNIT THICKNESS SLICE
L,1,5
L,2,3
L,101,105
L,102,103

```

4-16

```

LESIZE,ALL,,7
L,1,101
*REPEAT,5,1,1
LESIZE,5,,1
*REPEAT,5,1
LOCAL,11,1,,13
L,3,5
L,103,105
CSYS,1
MAT,2
MSHK,1                ! MAPPED VOLUME MESH
MSHA,0,3D              ! USING HEX
ESIZE,,4
V,1,2,3,5,101,102,103,105
V,5,3,4,4,105,103,104,104
VMESH,ALL
LOCAL,12,1,,23-1E-5,, -90 ! INTRODUCE SLIGHT INTERFERENCE
K,11                    ! CREATE UPPER SMALLER CYLINDER
K,12,10
K,13,10,8
K,14,10,90
K,15,8
KGEN,2,11,15,1,,,1,100
L,11,15
L,13,14
LESIZE,18,,,6
LESIZE,19,,,6
L,11,111
*REPEAT,5,1,1
LESIZE,20,,,1
*REPEAT,5,1
CSYS,11
L,13,15
L,113,115
CSYS,12
MAT,1
ESIZE,,6
V,12,13,15,15,112,113,115,115
V,15,13,14,11,115,113,114,111
VMESH,3,4
ASEL,S,AREA,,12
NSLA,,1                ! SELECT CONTACT NODES ON SMALLER CYLINDER
CM,CYL1,NODE           ! CONTACT NODES COMPONENT
REAL,1
TYPE,3
ESURF                  ! GENERATE 3-D CONTACT175 ELEMENTS
ASEL,S,AREA,,8
NSLA,,1                ! SELECT TARGET NODES ON BIGGER CYLINDER
CM,CYL2,NODE           ! TARGET NODES COMPONENT
REAL,1
TYPE,2
ESURF                  ! GENERATE 3-D TARGET170 ELEMENTS
NSEL,ALL
CSYS,0
NSEL,S,LOC,Y,23        ! SELECT TOP EDGE OF MODEL
CP,1,UY,ALL            ! COUPLE NODES ON TOP EDGE
*GET,NC,NODE,,NUM,MIN  ! GET LOWEST NODE NUMBER (MASTER)
NSEL,S,LOC,X           ! SYMMETRY CONSTRAINTS
D,ALL,UX
NSEL,S,LOC,Y
D,ALL,UY
NSEL,S,LOC,Z
D,ALL,UZ

```



```

NSEL,ALL
FINISH
SAVE,MODEL3D

*CREATE,SOLV3D,MAC      ! CREATE SOLUTION MACRO FOR 3-D CASE
/SOLU
D,NC,UY,-0.001          ! APPLY SMALL DISPLACEMENT TO ENGAGE CONTACT
SOLVE                   ! SOLVE FIRST LOAD STEP
DDELETE,NC,UY           ! DELETE IMPOSED DISPLACEMENT
F,NC,FY,-1600           ! APPLY HALF LOAD ON (SYMMETRY) MODEL
nsub,2,10,1
SOLVE                   ! SOLVE SECOND LOAD STEP
FINISH
*END

SOLV3D                  ! EXECUTE SOLUTION MACRO FOR 3-D CASE

*CREATE,RES3D,MAC       ! CREATE RESULTS MACRO FOR 3D CASE
/POST1
NSEL,,LOC,Y,23          ! SELECT TOP EDGE OF SMALLER CYLINDER
*GET,D,NODE,NC,U,Y      ! GET APPROACH DISTANCE (D)
ESEL,S,TYPE,,3          ! SELECT CONTACT ELEMENTS
ETABLE,NSTAT,CONT,STAT  ! STORE CONTACT STATUS
ESEL,R,ETAB,NSTAT,2,2   ! SELECT ELEMENTS WITH CONTACT (STAT=2)
CMSEL,S,CYL1            ! SELECT CONTACT COMPONENT NODES
NSLE,R                  ! RESELECT NODES WITH CONTACT
NSORT,LOC,X,1           ! SORT CONTACT NODES BY ASCENDING X LOCATION
*GET,B,SORT,,MAX        ! GET SEMI-CONTACT LENGTH (B)
*STATUS,PARM
*DIM,LABEL,CHAR,2,2
*DIM,VALUE,,2,3
LABEL(1,1) = 'AP DIS ', 'S-CON LEN '
LABEL(1,2) = ' mm', ' mm'
*VFILL,VALUE(1,1),DATA,-.4181,1.2
*VFILL,VALUE(1,2),DATA,D,B
*VFILL,VALUE(1,3),DATA,ABS(D/.4181),ABS(B/1.2)
FINISH
*END

RES3D                   ! EXECUTE POSTPROCESSING MACRO FOR 3-D CASE
SAVE,TABLE_2

/CLEAR, NOSTART
/COM
/COM  2-D ANALYSIS USING PLANE182 AND CONTA175
/COM  CONTACT ALGORITHM: LAGRANGE MULTIPLIER - KEYOPT(2) = 3
/COM
RESUME,MODEL2D
/PREP7
KEYOPT,3,2,3           ! LAGRANGE MULTIPLIER ON CONTACT NORMAL AND PENALTY ON
TANGENT
FINISH
SOLV2D
RES2D
SAVE,TABLE_3

/CLEAR, NOSTART
/COM
/COM  3-D ANALYSIS USING SOLID185 AND CONTA175
/COM  CONTACT ALGORITHM: LAGRANGE MULTIPLIER - KEYOPT(2) = 3
/COM
RESUME,MODEL3D
/PREP7

```

```

KEYOPT,3,2,3          ! LAGRANGE MULTIPLIER ON CONTACT NORMAL AND PENALTY ON
TANGENT
FINISH
SOLV3D
RES3D
SAVE, TABLE_4

RESUME, TABLE_1
/COM
/OUT, results, txt
/COM, ===== VM191 RESULTS COMPARISON =====
/COM,
/COM,                |   TARGET   |   ANSYS   |   RATIO
/COM,
/COM, 2-D ANALYSIS USING PLANE182 AND CONTAL75:
/COM, CONTACT ALGORITHM: AUGMENTED LAGRANGIAN
/COM,
*VWRITE, LABEL(1,1), LABEL(1,2), VALUE(1,1), VALUE(1,2), VALUE(1,3)
(1X, A8, A8, '      ', F10.4, '      ', F10.4, '      ', 1F5.3)
/NOPR
RESUME, TABLE_2
/GOPR
/COM,
/COM, 3-D ANALYSIS USING SOLID185 AND CONTAL75:
/COM, CONTACT ALGORITHM: AUGMENTED LAGRANGIAN
*VWRITE, LABEL(1,1), LABEL(1,2), VALUE(1,1), VALUE(1,2), VALUE(1,3)
(1X, A8, A8, '      ', F10.4, '      ', F10.4, '      ', 1F5.3)
/NOPR
RESUME, TABLE_3
/GOPR
/COM,
/COM, 2-D ANALYSIS USING PLANE182 AND CONTAL75:
/COM, CONTACT ALGORITHM: LAGRANGE MULTIPLIER
*VWRITE, LABEL(1,1), LABEL(1,2), VALUE(1,1), VALUE(1,2), VALUE(1,3)
(1X, A8, A8, '      ', F10.4, '      ', F10.4, '      ', 1F5.3)
/NOPR
RESUME, TABLE_4
/GOPR
/COM,
/COM, 3-D ANALYSIS USING SOLID185 AND CONTAL75:
/COM, CONTACT ALGORITHM: LAGRANGE MULTIPLIER
*VWRITE, LABEL(1,1), LABEL(1,2), VALUE(1,1), VALUE(1,2), VALUE(1,3)
(1X, A8, A8, '      ', F10.4, '      ', F10.4, '      ', 1F5.3)
/COM, =====
/OUT
FINISH

```



**P5.8** Deepdrawing is a manufacturing process that can create a complex shape out of a simply shaped plate (blank). The deepdrawing configuration is shown in the figure, which is composed of a blank, punch, die and blank holder. The thickness of the initial blank is 0.78mm. The die is fixed throughout the entire process, while the punch moves down by 30mm to shape the blank. The holder controls the slip of the blank by applying friction force. The fillet radii of both punch and die are 5mm. After the maximum down-stroke of the punch, both the punch and holder are removed. Then, the blank will experience elastic springback. The objective of this project is to simulate the final geometry of the blank after springback.

Model the process using an axi-symmetric problem. You may use CAX4R elements. The whole simulation is divided by three steps. (1) The blank holder is pushed (displacement control) to provide about 100kN of holding force. (2) While the blank holder is fixed at the location of step (1), the punch is moved down by 30mm. (3) Punch, die, and blank holder are removed so that the blank is elastically deformed by springback. It is possible to change processes.

The following results need to be submitted: (1) deformed shape plots of five different steps, (2) graph of radial position vs. radial strain, and (3) graph of radial position vs. thickness change, (4) graph of punch displacement vs. punch force, and (5) comparison of deformed shapes at the maximum stroke and after springback.




Figure P5.8 Deepdrawing problem

### Solution:

The geometry of the problem is shown in Figure P5.8. The circular blank being drawn has an initial radius of 100 mm and an initial thickness of 0.78 mm. The punch has a radius of 25 mm and is rounded off at the corner with a radius of  $u_3 = 13 \text{ mm}$ . The die has an internal radius of 56 mm and is rounded off at the corner with a radius of  $u_4 = 5 \text{ mm}$ . The blank holder has an internal radius of 26 mm.

The blank is modeled using 40 elements of type CAX4R. The contact between the blank and the rigid punch, the rigid die, and the rigid blank holder is modeled with the \*CONTACT PAIR option. The top and bottom surfaces of the blank are defined by means of the \*SURFACE option. The rigid punch, the die, and the blank holder are modeled as analytical rigid surfaces with the \*RIGID BODY option in conjunction with the \*SURFACE option. The mechanical interaction between the contact surfaces is assumed to be frictional contact. Therefore, the \*FRICTION option is used in conjunction with the various \*SURFACE INTERACTION property options to specify coefficients of friction.

At the start of the analysis, the blank is positioned precisely on top of the die and the blank holder is precisely in touch with the top surface of the blank. The punch is positioned 0.18 mm above the top surface of the blank.

The entire analysis is carried out in five steps. In the first step the blank holder is pushed onto the blank with a prescribed displacement to establish contact. In the shell models this displacement roughly corresponds to zero clearance across the interface.


In the second step the boundary condition is removed and replaced by the applied force of 100 kN on the blank holder. This force is kept constant during Steps 2 and 3. This technique of simulating the clamping process is used to avoid potential problems

with rigid body modes of the blank holder, since there is no firm contact between the blank holder, the blank, and the die at the start of the process. The two-step procedure creates contact before the blank holder is allowed to move freely.


In the third step the punch is moved toward the blank through a total distance of 30 mm. This step models the actual drawing process. During this step the option \*CONTROLS, ANALYSIS=DISCONTINUOUS is included since contact with friction tends to create a severely discontinuous nonlinearity and we wish to avoid premature cutbacks of the automatic time incrementation scheme.

The last two steps are used to simulate springback. In the fourth step all the nodes in the model are fixed in their current positions and the contact pairs are removed from the model with the \*MODEL CHANGE, TYPE=CONTACT PAIR, REMOVE option. This is the most reliable method for releasing contact conditions. In the fifth, and final, step the regular set of boundary conditions is reinstated and the springback is allowed to take place. This part of the analysis with the CAX4R elements is included to demonstrate the feasibility of the unloading procedure only and is not expected to produce realistic results, since the reduced-integration elements have a purely elastic bending behavior. The springback is modeled with more accuracy in the shell element models.


The following figure shows deformed shapes that are predicted at various stages of the drawing process. The profiles show that the metal initially bends and stretches and is then drawn in over the surface of the die.



The distributions of radial and circumferential strain for all three models and thickness strain are shown in the following figure. The thickness does not change very much: the change ranges from approximately  $-12\%$  in the cylindrical part to approximately  $+16\%$  at the edge of the formed cup. Relatively small thickness changes are usually desired in deep drawing processes and are achieved because the radial tensile strain and the circumferential compressive strain balance each other.




The drawing force as a function of punch displacement is shown in the following figure, where the results are compared with different types of element, and all curves are seen to match closely.



The deformed shape after complete unloading is shown in the following figure, superimposed on the deformed shape under complete loading. The analysis shows the lip of the cup springing back strongly after the blank holder is removed. The springback in the CAX4R model is not physically realistic: in the first-order reduced-integration elements an elastic “hourglass control” stiffness is associated with the “bending” mode, since this mode is identical to the “hourglass” mode exhibited by this element in continuum situations. In reality the bending of the element is an elastic-plastic process, so that the springback is likely to be much less. A better simulation of this aspect would be achieved by using several elements through the thickness of the blank, which would also increase the cost of the analysis. The springback results for the shell models do not exhibit this problem and are clearly more representative of the actual elastic-plastic process.

CAX4R MODEL



The following program is the ANSYS command script to solve the problem:

```

*HEADING
  DEEP DRAWING OF CYLINDRICAL CUP WITH CAX4R
*RESTART,WRITE,FREQUENCY=25
*NODE
101,0.0,0.0
181,100.,0.0
301,0.0,0.78
381,100.,0.78
*NGEN,NSET=BOT
101,181,2
*NGEN,NSET=TOP
301,381,2
*NSET,NSET=WRKPC
BOT, TOP
*NODE,NSET=DIE
100,0.1,-50.
*NODE,NSET=PUNCH
200,0.,50.
*NODE,NSET=HOLDER
300,0.1,50.
*NSET,NSET=TOOLS
PUNCH,DIE,HOLDER
*NSET,NSET=CENTER
101,301
*ELEMENT,TYPE=CAX4R,ELSET=BLANK
201,101,103,303,301
*ELGEN,ELSET=BLANK
201,40,2,2
*ELSET,ELSET=ALL
BLANK,
*SOLID SECTION,MATERIAL=STEEL,ORIENTATION=LOCAL,
ELSET=BLANK
*ORIENTATION,NAME=LOCAL
1.,0.,0.,0.,1.,0.
0,0.,
*MATERIAL,NAME=STEEL
*ELASTIC
2.06E5,0.29
*PLASTIC,HARDENING=ISOTROPIC
  0.167E+03, 0.00000E+00
  0.425E+03, 2.0
*ELSET,ELSET=EDIE,GENERATE
231,279,2
*ELSET,ELSET=EHOLDER,GENERATE
241,279,2
*ELSET,ELSET=EPUNCH,GENERATE

```

```

201,249,2
*RIGID BODY,ANALYTICAL SURFACE=BSURF,REF NODE=100
*SURFACE,TYPE=SEGMENTS,NAME=BSURF
START,26,-60.
LINE,26,-5.
CIRCL,31,0.,31,-5.
LINE,100,0.
*RIGID BODY,ANALYTICAL SURFACE=DSURF,REF NODE=300
*SURFACE,TYPE=SEGMENTS,NAME=DSURF
START,100,0.78
LINE,31,0.78
CIRCL,26,5.78,31,5.78
LINE,26,60.78
*RIGID BODY,ANALYTICAL SURFACE=FSURF,REF NODE=200
*SURFACE,TYPE=SEGMENTS,FILLET RADIUS=.013,
NAME=FSURF
START,25,60.78
LINE,25,5.78
CIRCL,20,0.78,20,5.78
LINE,0.0,0.78
*SURFACE,NAME=ASURF
EDIE,S1
*SURFACE,NAME=CSURF
EHOLDER,S3
*SURFACE,NAME=ESURF
EPUNCH,S3
*CONTACT PAIR,INTERACTION=ROUGH1
ASURF,BSURF
*CONTACT PAIR,INTERACTION=ROUGH2
CSURF,DSURF
*CONTACT PAIR,INTERACTION=ROUGH3
ESURF,FSURF
*SURFACE INTERACTION,NAME=ROUGH1
*FRICTION
0.144,
*SURFACE INTERACTION,NAME=ROUGH2
*FRICTION
0.144,
*SURFACE INTERACTION,NAME=ROUGH3
*FRICTION
0.144,
*STEP,NLGEOM, UNSYMM=YES
PUSH THE BLANKHOLDER DOWN BY A PRESCRIBED
DISPLACEMENT
*STATIC
1.,1.
*BOUNDARY
CENTER,1,1
DIE,1,1
DIE,2,2
DIE,6,6
PUNCH,1,1
PUNCH,2,2
PUNCH,6,6
HOLDER,1,1
HOLDER,2,2,-1.75E-5
HOLDER,6,6
*MONITOR,NODE=200,DOF=2
*CONTACT CONTROLS,FRICTION ONSET=DELAY
*PRINT,CONTACT=YES
*NODE PRINT,NSET=TOOLS,FREQUENCY=100
COORD,U,RF
*EL PRINT,ELSET=ALL,FREQUENCY=500

```

```

S,E
*NODE FILE,NSET=TOOLS,FREQUENCY=10
U,RF
*CONTACT FILE,SLAVE=ASURF,FREQUENCY=10
*CONTACT FILE,SLAVE=CSURF,FREQUENCY=10
*CONTACT FILE,SLAVE=ESURF,FREQUENCY=10
*END STEP
*STEP,NLGEOM
APPLY PRESCRIBED FORCE ON BLANKHOLDER AND
RELEASE DISPLACEMENT
*STATIC
1.,1.
*BOUNDARY,OP=NEW
CENTER,1,1
DIE,1,1
DIE,2,2
DIE,6,6
PUNCH,1,1
PUNCH,2,2
PUNCH,6,6
HOLDER,1,1
HOLDER,6,6
*CLOAD
HOLDER,2,-1000.
*END STEP
*STEP,INC=500,NLGEOM
MOVE THE PUNCH DOWN
*STATIC
.01,1.,1.E-6
*CONTROLS,ANALYSIS=DISCONTINUOUS
*BOUNDARY,OP=NEW
CENTER,1,1
DIE,1,1
DIE,2,2
DIE,6,6
PUNCH,1,1
PUNCH,2,2,-30.0
PUNCH,6,6
HOLDER,1,1
HOLDER,6,6
*CLOAD
HOLDER,2,-1000.
*END STEP
*STEP,NLGEOM
FIX ALL NODES AND REMOVE THE CONTACT SURFACES
*STATIC
1.,1.,1.,1.
*BOUNDARY,FIXED
WRKPC,1,2
*MODEL CHANGE,TYPE=CONTACT PAIR,REMOVE
ASURF,BSURF
CSURF,DSURF
ESURF,FSURF
*CLOAD,OP=NEW
HOLDER,2,0.
*END STEP
*STEP,NLGEOM, UNSYMM=NO
REPLACE THE BOUNDARY CONDITIONS BY THE REGULAR SET
*STATIC
.1,1.,1.E-6
*BOUNDARY,OP=NEW
181,2
CENTER,1,1

```



\*END STEP

